

Template for *Computer Project 3*

We want to demonstrate the method of Lagrange Multipliers by finding critical points for the function

$$f(x, y, z) = xz + y^2$$

on the surface

$$x^2 + y^2 + 3z^2 = 1.$$

First, we use *Mathematica* to find the critical points according to the method. Since the points for your given functions may be fairly nasty, it's OK to use `NSolve!`

```
In[1]:= f[x_, y_, z_] := x * z + y^2
g[x_, y_, z_] := x^2 + y^2 + 3 z^2 - 1
```

```
In[3]:= gradf = Grad[f[x, y, z], {x, y, z}]
gradg = Grad[g[x, y, z], {x, y, z}]
```

```
Out[3]= {z, 2 y, x}
```

```
Out[4]= {2 x, 2 y, 6 z}
```

```
In[5]:= soln = NSolve[{gradf == λ * gradg, g[x, y, z] == 0}, {x, y, z, λ}]
```

```
Out[5]= {{x → 0., y → 1., z → 0., λ → 1.}, {x → -0.707107, y → 0., z → 0.408248, λ → -0.288675},
{x → 0.707107, y → 0., z → -0.408248, λ → -0.288675},
{x → 0.707107, y → 0., z → 0.408248, λ → 0.288675}, {x → 0., y → -1., z → 0., λ → 1.},
{x → -0.707107, y → 0., z → -0.408248, λ → 0.288675}}
```

So, we have 6 critical points in all. First, we prepare a table that substitutes these points into the objective function $f(x, y, z)$. Note that the `/.` syntax takes each set of replacement rules above and substitutes the values for x , y , and z .


```
In[6]:= Grid[Prepend[Table[{x, y, z, f[x, y, z]} /. soln[[i]], {i, 1, Length[soln]}],
{"x", "y", "z", "f(x,y,z)"}], Frame → All]
```

```
Out[6]=
```

| x | y | z | f(x,y,z) |
|-----------|-----|-----------|-----------|
| 0. | 1. | 0. | 1. |
| -0.707107 | 0. | 0.408248 | -0.288675 |
| 0.707107 | 0. | -0.408248 | -0.288675 |
| 0.707107 | 0. | 0.408248 | 0.288675 |
| 0. | -1. | 0. | 1. |
| -0.707107 | 0. | -0.408248 | 0.288675 |

We can easily see that the absolute maximum value of f over the ellipsoid is 1 which occurs at $(0,1,0)$ and $(0,-1,0)$. The absolute minimum is -0.288675 which occurs at the points $(0.707107, 0, -0.408248)$ and $(-0.707107, 0, 0.408248)$. Next, we use `ContourPlot3D` to get a graph of our surface. We can use the `ColorFunction` option to color the surface according to our objective function $f(x, y, z)$.


```
In[7]:= p1 = ContourPlot3D[g[x, y, z] == 0, {x, -1.1, 1.1}, {y, -1.1, 1.1}, {z, -1.1, 1.1},  
  Mesh → None, ColorFunctionScaling → False, Lighting → None,  
  PlotPoints → 100, ColorFunction → Function[{x, y, z}, Glow[Hue[f[x, y, z]]],  
  AxesLabel → {"x", "y", "z"}]
```



```
In[8]:=
```

We now add ALL of our critical points as small spheres. You may need to adjust the radius of the spheres to see them clearly.

```
In[9]:= p2 = Graphics3D[Table[Sphere[{x, y, z}, 0.025] /. soln[[i]], {i, 1, Length[soln]}];  
Show[p1, p2]
```



As you can see, some of the points we found look like they sit in the middle of valleys or mountains (low or high values of f). Others look like they sit at saddle points (more of an X-shape in the color).